

Deductive Databases
Summer Term 2019
 Prof. Dr. W. May

3. Unit: Well-founded and Stable Semantics

Exercise 1 (Well-Founded Model) a) Show that there are non-stratifiable Datalog[¬] programs that have a total well-founded model (i.e., no atoms undefined).

b) Are there (non-ground) non-stratifiable Datalog[¬] programs that have a total well-founded model for *all* EDB instances?

a) Take a simple win-move game that has only won and lost positions, no drawn ones:

```
pos(a). pos(b). pos(c).
move(a,b).
move(b,c).
win(X) :- move(X,Y), not win(Y).
```

The well-founded model is

({pos(a). pos(b). pos(c). move(a,b). move(b,c). win(b)},
 {move(a,a). move(a,c). move(b,a). move(b,a). move(c,a). move(c,b). move(c,c). win(a). win(c).})


b) Consider EDB relations $p/1, q/1, s_0/1, t_0/1$. The program P is as follows:

```
r(x) :- p(x), not q(x).
s(x) :- s0(x).
s(x) :- q(x), not t(x).
t(x) :- t0(x).
t(x) :- r(x), not s(x).
```

Sketch: The program describes a partition that is based on splitting p into q vs. r . $p \wedge q$ is one side, $p \wedge \neg q$ the other.

Based on this, relations s vs. t are defined (which are not necessarily disjoint): By “default”, elements of q belong to s , while elements of r belong to t . The membership of elements can be influenced by s_0 and t_0 that “overwrites” the above defaults, which is encoded into the $q \rightarrow s$ and $r \rightarrow t$ rules that create a negative cyclic dependency. (Note that elements a can be assigned to be both in s and t via $s_0(a)$ and $t_0(a)$).


The dependency graph is



For each EDB instance that defines $I(p), I(q), I(s_0), I(t_0)$, the well-founded model is total.

Exercise 2 (Well-Founded Model) Give an instance of the win-move game that has no total stable model.

Cycle with three positions:



```

win(X) :- move(X,Y), not win(Y).
lose(X) :- pos(X), not win(X).
pos(a).
pos(b).
pos(c).
move(a,b).
move(b,c).
move(c,a).
% lparse -n 0 -d none --partial winmovenontotal1.s |smodels

```

[Filename: winmovenontotal1.s]

The only stable model is M with

$$\begin{aligned}
 val_M(win(a)) &= val_M(win(b)) = val_M(win(c)) = u, \\
 val_M(lose(a)) &= val_M(lose(b)) = val_M(lose(c)) = u.
 \end{aligned}$$

In general: any cycle with an odd number of positions, and where no position is lost due to an exit from the cycle.



```

win(X) :- move(X,Y), not win(Y).
lose(X) :- pos(X), not win(X).
pos(a1).
pos(b1).
pos(c1).
pos(d1).
move(a1,b1).
move(b1,c1).
move(c1,a1).
move(c1,d1).

pos(a2).
pos(b2).
pos(c2).
pos(d2).
pos(e2).
move(a2,b2).
move(b2,c2).

```

```

move(c2,a2).
move(c2,d2).
move(d2,e2).
% lparse -n 0 -d none --partial winmovenontotal2.s |smodels

```

[Filename: winmovenontotal2.s]

In the “1” game, the exit makes d_1 a losing position and thus c_1 is a winning position (move to d_1). Thus, b_1 is lost and a_1 is won.

In the “2” game, the exit makes e_1 lost and d_1 won, but c_1 is not lost, since he player will move to a_1 and stay in the cycle.


Note: a minimal such example ist a win-move game with only a single node and move:

```

win(X) :- move(X,Y), not win(Y).
pos(a).
move(a,a).

```

Exercise 3 (Well-Founded Model) Consider again the win-move game from the lecture:



Consider to start the Alternating Fixpoint Computation for the rules $\text{win}(X) :- \text{move}(X,Y), \text{not win}(Y)$,

$\text{lose}(X) :- \text{pos}(X), \text{not win}(X)$.

with \mathcal{H}_0 as

- some atoms that are correct: $\text{lose}(k), \text{win}(b), \text{win}(d)$
- some atoms that actually are in contrast to the well-founded model of the above game: $\text{win}(f), \text{lose}(c), \text{win}(m)$.

(it is often called “seed” when starting an iterative algorithm with some initial values)

-
- Note: lose is actually not used in any rule body; interpreting lose as $\neg \text{win}$ does also not contain any information for the seed, since $\neg \text{win}$ is assumed to hold by default.
→ in general, any negative seed does not have any consequence.
 - So the seed is mainly an underestimate, and only the positive atoms add some flavour of overestimate.

The first reduct, $P_{\mathcal{H}_0}$ contains the following rules (all rules vor pairs x, y s.t. there is no move from x to y are omitted):

win(a) : -move(a, f), win(f).	lose(a) : - win(a) true.
win(a) : -move(a, b), win(b).	lose(b) : -win(b).
win(b) : -move(b, c), win(c).	lose(c) : - win(c) true.
win(b) : -move(b, k), win(k).	lose(d) : -win(d).
win(c) : -move(c, d), win(d).	lose(e) : - win(e) true.
win(c) : -move(c, l), win(l).	lose(f) : -win(f).
win(l) : -move(l, d), win(d).	lose(g) : - win(g) true.
win(d) : -move(d, e), win(e).	lose(h) : - win(h) true.
win(e) : -move(e, a), win(a).	lose(i) : - win(i) true.
win(i) : -move(i, j), win(j).	lose(j) : - win(j) true.
win(g) : -move(g, i), win(i).	lose(k) : - win(k) true.
win(b) : -move(b, g), win(g).	lose(l) : - win(l) true.
win(g) : -move(g, h), win(h).	lose(m) : -win(m).
win(h) : -move(h, m), win(m).	lose(n) : - win(n) true.
win(m) : -move(m, h), win(h).	

Thus, $\mathcal{H}_1 = T_{P_0}^\omega(\emptyset) = \{ \text{win}(b), \text{win}(c), \text{win}(d), \text{win}(e), \text{win}(g), \text{win}(i), \text{win}(m),$
 $\text{lose}(a), \text{lose}(c), \text{lose}(e), \text{lose}(g), \text{lose}(h), \text{lose}(i), \text{lose}(j), \text{lose}(k), \text{lose}(l), \text{lose}(n) \}$

Note: Usually, \mathcal{H}_1 would be an overestimate, but win(a) is (still) missing, since f was wrongly set to be won. Note that now, f is not re-derived to be won (i.e. it is already clear that it is not supported), but also lose(f) is not yet there.

win(h) is not there, since m was fixed to be won (both are drawn in the original game).

The second reduct, $P_{\mathcal{H}_1}$ contains the following rules (again, rules with no underlying move are omitted):

win(a) : -move(a, f), win(f).	lose(a) : - win(a) true.
win(a) : -move(a, b), win(b).	lose(b) : -win(b).
win(b) : -move(b, c), win(c).	lose(c) : -win(c).
win(b) : -move(b, k), win(k).	lose(d) : -win(d).
win(c) : -move(c, d), win(d).	lose(e) : -win(e).
win(c) : -move(c, l), win(l).	lose(f) : - win(f) true.
win(l) : -move(l, d), win(d).	lose(g) : -win(g).
win(d) : -move(d, e), win(e).	lose(h) : - win(h) true.
win(e) : -move(e, a), win(a).	lose(i) : -win(i).
win(i) : -move(i, j), win(j).	lose(j) : - win(j) true.
win(g) : -move(g, i), win(i).	lose(k) : - win(k) true.
win(b) : -move(b, g), win(g).	lose(l) : - win(l) true.
win(g) : -move(g, h), win(h).	lose(m) : -win(m).
win(h) : -move(h, m), win(m).	lose(n) : - win(n) true.
win(m) : -move(m, h), win(h).	

Thus, $\mathcal{H}_2 = T_{P_{\mathcal{H}_1}}^\omega(\emptyset) = \{ \text{win}(a), \text{win}(b), \text{win}(c), \text{win}(e), \text{win}(g), \text{win}(i), \text{win}(m),$
 $\text{lose}(a), \text{lose}(f), \text{lose}(h), \text{lose}(j), \text{lose}(k), \text{lose}(l), \text{lose}(n) \}$

Note: Usually, \mathcal{H}_2 would be an underestimate. win(a) and lose(f) are now there, thus, the f issue is already corrected. But, win(e) is there, which is a consequence of having lose(a) before ... the f issue propagates through the game now.

Note that win(d) is not there, although it was there in the seed (an is finally true), but not yet supported.

win(c) is there, lose(c) is not there, so the c issue is also corrected (if there would be a longer chain between c and b, it would propagate through it until finally dying).

lose(h) and win(m) are both there (both are drawn in the original game, but win(m) was fixed in the seed).

The third reduct, $P_{\mathcal{H}_2}$ contains the following rules (again, rules with no underlying move are omitted):

- | | |
|---|---|
| $win(a) : -move(a, f), win(f).$ | $lose(a) : -win(a).$ |
| $win(a) : -move(a, b), win(b).$ | $lose(b) : -win(b).$ |
| $win(b) : -move(b, c), win(c).$ | $lose(c) : -win(c).$ |
| $win(b) : -move(b, k), win(k).$ | $lose(d) : -win(d) true.$ |
| $win(c) : -move(c, d), win(d).$ | $lose(e) : -win(e).$ |
| $win(c) : -move(c, l), win(l).$ | $lose(f) : -win(f) true.$ |
| $win(l) : -move(l, d), win(d).$ | $lose(g) : -win(g).$ |
| $win(d) : -move(d, e), win(e).$ | $lose(h) : -win(h) true.$ |
| $win(e) : -move(e, a), win(a).$ | $lose(i) : -win(i).$ |
| $win(i) : -move(i, j), win(j).$ | $lose(j) : -win(j) true.$ |
| $win(g) : -move(g, j), win(j).$ | $lose(k) : -win(k) true.$ |
| $win(b) : -move(b, g), win(g).$ | $lose(l) : -win(l) true.$ |
| $win(g) : -move(g, h), win(h).$ | $lose(m) : -win(m).$ |
| $win(h) : -move(h, m), win(m).$ | $lose(n) : -win(n) true.$ |
| $win(m) : -move(m, h), win(h).$ | |

Thus, $\mathcal{H}_3 = T_{P_{\mathcal{H}_2}}^\omega(\emptyset) = \{ win(a), win(b), win(c), win(g), win(i), win(l), win(m), lose(d), lose(f), lose(h), lose(j), lose(k), lose(l), lose(n) \}$


\mathcal{H}_3 would be an overestimate, but currently contains neither win(e) nor lose(e). This is still a consequence of the f issue.


On the other hand, the sequence shows that g, h, i, j, k, l, m and n do not change any more. One can also know that a (win because of f), b (win because of k) and f (lose) will not change any more.

The rest follows and corrects then from the usual well-founded characteristics of the AFP computation.

The process can also be visualized by the original game (Green= win, recall that the lose predicate has been introduced only for demonstrational issues):

(“win: can move to a non-win node”)





$\mathcal{H}_7 = \mathcal{H}_6$, is one of the two total stable models, namely the one which makes the seeded $\text{win}(m)$ true.

Conclusions:

- “Seeding” correct positive atoms does not help much if the seed does not contain their support (recall: checking stability of the well-founded model is actually a full seed and reproduces in a single step!).

- “Seeding” wrong positive atoms will remove them since they are not supported.
- “Seeding” positive atoms where the well-founded model is undefined can result in a stable model (if the support for there stability is also contained in the seed).
- “Seeding” positive atoms in general will not result in a fixpoint (seeding into a 3-cycle, where no total stable model exists; or only one win-position into a four-cycle, which is not sufficient to reproduce).

Note that it does in general *not* extend the interpretation to a stable model. Nevertheless, this can be used for strategies to find stable models.

Exercise 4 (Stable Models: soccer league) A newspaper article (in german) took the ranking of the 2nd german soccer league (after 29 rounds of the 2017/18 season, Friday april 13th morning) and shows a possible final table (after 34 rounds) where all teams from the 4th to the last, 18th place have 44 points each (the ranking is actually quite dense, all teams from 5th position on fight against relegation).

Original article:

<https://www.welt.de/sport/article175403621/Fussball-Die-wahnsinnige-Tabellensituation-in-der-Zweiten.html>

How must the teams play such that this final table would come true?

Use the following fragment which contains all relevant data about the season:

```
round(29..34).
p(29..68).    %% maximally 68P can be reached by the current leader

team(d).    team(n).    team(ki).    team(r).    team(in).    team(bo).
team(bi).    team(aue).    team(svs).    team(du).    team(un).    team(bs).
team(dd).    team(sp).    team(ft).    team(hdh).    team(da).    team(kl).

% standings after 29 rounds:
points(29,d,53).
points(29,n,50).
points(29,ki,46).
points(29,r,41).
points(29,in,41).
points(29,bo,40).
points(29,bi,40).
points(29,aue,39).
points(29,svs,38).
points(29,du,38).
points(29,un,37).
points(29,bs,37).
points(29,dd,37).
points(29,sp,37).
points(29,ft,37).
points(29,hdh,34).
points(29,da,32).
points(29,kl,29).

% games of the remaining rounds:
game(30,in,n).    game(30,hdh,d).    game(30,dd,ki).
game(30,du,svs).    game(30,ft,r).    game(30,sp,un).
```

```

game(30,bi,aue).   game(30,da,bs).   game(30,bo,kl).

game(31,ki,n).   game(31,d,in).   game(31,aue,du).
game(31,r,sp).   game(31,svs,da). game(31,un,hdh).
game(31,bs,bi).   game(31,kl,dd).   game(31,ft,bo).

game(32,n,bs).   game(32,dd,d).   game(32,in,ki).
game(32,du,r).   game(32,hdh,svs). game(32,da,un).
game(32,bi,kl).   game(32,sp,ft).   game(32,bo,aue).

game(33,svs,n).   game(33,d,ki).   game(33,ft,du).
game(33,r,da).   game(33,un,bo).   game(33,sp,bi).
game(33,bs,in).   game(33,kl,hdh).   game(33,aue,dd).

game(34,n,d).   game(34,ki,bs).   game(34,du,sp).
game(34,bo,r).   game(34,dd,un).   game(34,bi,svs).
game(34,in,kl).   game(34,hdh,ft).   game(34,da,aue).

```

Guess, how many possibilities exist.

The problem can be solved by 6+1 rules:

- every game is either won by the home team or by the guest team, or drawn,
- 4 rules for updating the points of the teams,
- a rule that specifies that the number of points at the end must be unique,
- and the requirement that the 15 teams each have 44 points at the end.

The solution lists 445878 possible models!

```

round(29..34).
p(29..68).   %% maximally 68P can be reached by the current leader

team(d).   team(n).   team(ki).   team(r).   team(in).   team(bo).
team(bi).   team(aue).   team(svs).   team(du).   team(un).   team(bs).
team(dd).   team(sp).   team(ft).   team(hdh).   team(da).   team(kl).

% standings after 29 rounds:
points(29,d,53).
points(29,n,50).
points(29,ki,46).
points(29,r,41).
points(29,in,41).
points(29,bo,40).
points(29,bi,40).
points(29,aue,39).
points(29,svs,38).
points(29,du,38).
points(29,un,37).
points(29,bs,37).
points(29,dd,37).
points(29,sp,37).
points(29,ft,37).

```



```

points(29,hdh,34).
points(29,da,32).
points(29,kl,29).

% games of the remaining rounds:
game(30,in,n).   game(30,hdh,d).   game(30,dd,ki).
game(30,du,svs). game(30,ft,r).   game(30,sp,un).
game(30,bi,aue). game(30,da,bs).   game(30,bo,kl).

game(31,ki,n).   game(31,d,in).   game(31,aue,du).
game(31,r,sp).   game(31,svs,da). game(31,un,hdh).
game(31,bs,bi).  game(31,kl,dd).  game(31,ft,bo).

game(32,n,bs).   game(32,dd,d).   game(32,in,ki).
game(32,du,r).   game(32,hdh,svs). game(32,da,un).
game(32,bi,kl).  game(32,sp,ft).  game(32,bo,aue).

game(33,svs,n).  game(33,d,ki).   game(33,ft,du).
game(33,r,da).   game(33,un,bo).  game(33,sp,bi).
game(33,bs,in).  game(33,kl,hdh). game(33,aue,dd).

game(34,n,d).    game(34,ki,bs).  game(34,du,sp).
game(34,bo,r).   game(34,dd,un).  game(34,bi,svs).
game(34,in,kl).  game(34,hdh,ft). game(34,da,aue).

1{ win(R,X,Y), draw(R,X,Y), win(R,Y,X) }1 :- game(R,X,Y).
draw(R,Y,X) :- draw(R,X,Y), game(R,X,Y).
draw(R,Y,X) :- draw(R,X,Y), game(R,Y,X).
%%% "draw" symmetric makes it easier to "seed" results.

points(RR,X,M) :- win(RR,X,Y),  points(R,X,N), RR = R+1, M = N+3, round(R), team(X), team(Y), p(N).
points(RR,Y,N) :- win(RR,X,Y),  points(R,Y,N), RR = R+1,          round(R), team(X), team(Y), p(N).
points(RR,X,M) :- draw(RR,X,Y), points(R,X,N), RR = R+1, M = N+1, round(R), team(X), team(Y), p(N).

%%% Alternative, less code, but slower
% 1{ result(R,X,Y,3,0), result(R,X,Y,1,1), result(R,X,Y,0,3) }1 :- game(R,X,Y).
% points(RR,X,PX2) :- result(RR,X,Y,DX,DY), points(R,X,PX),
%                   RR = R+1, PX2 = PX+DX, round(R), team(X), team(Y), p(PX), p(PY), e(DX,DY).
% points(RR,Y,PY2) :- result(RR,X,Y,DX,DY), points(R,Y,PY),
%                   RR = R+1, PY2 = PY+DY, round(R), team(X), team(Y), p(PX), p(PY), e(DX,DY).

points(34,X,44) :- team(X), X != n, X != ki, X != d.
:- points(34,X,N1), points(34,X,N2), team(X), p(N1), p(N2), N1 != N2.

% kill models where N and D have more points than KI (113247 remain)
% :- points(34,n,PN), points(34,ki,PK), points(34,d,PD), PN > PK, PD > PK, p(PN), p(PD), p(PK).

% Can N still end up on 3rd place then?
% kill models where KI and N do not have more points than D: (80926 remain)
% :- points(34,n,PN), points(34,ki,PK), PN > PK, p(PN), p(PK).
% :- points(34,n,PN), points(34,d,PD), PN > PD, p(PN), p(PD).

```

```
% kill models there N, D, and KI do not have the same number of points (1877 models remain)
%:- points(34,n,P1), points(34,d,P2), P1 != P2, p(P1), p(P2).
%:- points(34,n,P1), points(34,ki,P2), P1 != P2, p(P1), p(P2).
%:- points(34,d,P1), points(34,ki,P2), P1 != P2, p(P1), p(P2).
% points(34,ki,58).    %% all have 58P: 295, 57P: 756, 56P: 585, 55P: 241
```
